
1 Overview

This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2019-06_D20wk25 related to the title of this CR.

Contact:          Fu Wang
wangfu@datangmobile.cn
2 Corrections required
2.1 Change 1

	Function name
	fl_TC_6_3_1_1_TestBody ()

	Reason for change
	1. According to TS 31.102,

If service n°127 is "available" in the USIM Service Table, the Registration Accept shall contain control plane based Steering of Roaming information during the initial registration procedure in a VPLMN. 

So Steering of Roaming information shall be configured in the Registration Accept during the initial registration procedure in a VPLMN at step step22a1 - 22a16a1.
Note : it will have a prose CR to support.


	Summary of change
	1. Rewrite step22a1 - 22a16a1.

	TTCN module
	SOR_NR5GC.ttcn

	MCC160 Comment
	Accepted in principle, but implemented differently.


Before 

function fl_TC_6_3_1_1_TestBody() runs on NR5GC_PTC

  {

    var NR_SRB_COMMON_IND v_ReceivedAsp;

    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;//@sic R5-202596 sic@

    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;//@sic R5-202596 sic@

    var TypeOfRegistration_Type v_TypeOfRegistration;//@sic R5-202596 sic@

    var O16_Type v_Mac_UE;

    var NAS_PlmnId v_PlmnID_2 := f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (nr_Cell11));

    var O2_Type v_AccessTechId :='0800'O;//@sic R5s200663 sic@

    var octetstring v_Data := v_PlmnID_2 & v_AccessTechId;

    var template (value) SOR_Header v_Header := crs_NG_SOR_Header('1'B,'1'B,'1'B,'0'B);

    timer t_Tmax;

    //@siclog "Step 1" siclog@

    //Power on the UE

    f_UT_SwitchOnUE(UT);

    //@siclog "Steps 2-14" siclog@

    //Steps 1 to 13 of the registration procedure described in TS 38.508 subclause 4.5.2.2-2 are performed on NR Cell 13

    //@siclog "Step 15" siclog@

    //The SS transmits an DLInformationTransfer message and an REGISTRATION ACCEPT message containing

    //steering of roaming information indicating list of preferred PLMN/access technology combination provided with acknowledgment requested from the UE for successful reception

    v_Mac_UE:=f_NR5GC_SorRegistration(nr_Cell13,v_Data,v_Header);

    //@siclog "Step 16" siclog@

    //SS starts timer of tmax =(6 minutes + cell selection time) (Note 1, 2 and 3)

    t_Tmax.start(f_NR_SetTimerToleranceMax(nr_Cell13, nonProtocolTimer, 660.0));

    //@siclog "Step 17" siclog@

    //The UE transmits an ULInformationTransfer message and REGISTRATION COMPLETE message carrying acknowledgement of successful reception of the steering of roaming information

    SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell13, tsc_NR_RbId_SRB1, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

    if (not match (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg, cr_NG_REGISTRATION_COMPLETE(cr_NG_SORTransparentContainer(crs_NG_SOR_Header('0'B,'0'B,'0'B,'1'B),v_Mac_UE)))) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Complete Message Failed");

      }

    //@siclog "Steps 18a1-18a5a1" siclog@

    //Step 19a1 of the registration procedure described in TS 38.508-1 [4] Table 4.5.2.2-2 is performed.

    //@siclog "Step 18" siclog@

    //The SS transmits an RRCRelease message.

    f_NR5GC_RRC_Idle_Steps16_20(nr_Cell13,-,TESTMode_OFF,rrcConnectionRelease);//@sic R5-202596 sic@

    //@siclog "Step 19" siclog@

    //Check: Does the UE transmits an RRCSetupRequest on NR Cell 11 before tmax expires?

    alt {

      [] SRB.check(receive(car_NR_SRB0_RrcPdu_IND(nr_Cell11, cr_38508_RRCSetupRequest(cr_NR_EstablishmentCause_Any))))

        {

          t_Tmax.stop;

          f_NR_PreliminaryPass (__FILE__, __LINE__, "Step 19 PASS");

        }

      [] t_Tmax.timeout

        {

          f_NR_SetVerdictFailOrInconc (__FILE__, __LINE__, "Step 19 Failed");

        }

    }

    //@siclog "Step 20-21" siclog@

    //Steps 3 to 4 of the registration procedure described in TS 38.508-1 subclause 4.5.2.2-2 are performed on NR Cell 11.

    //@sic R5-202596 sic@

    v_TypeOfRegistration := f_NR5GC_InitialOrMobilityRegistrationRequest (v_GMM_MobilityInfo, v_ReceivedMsg, nr_Cell11);

    if (v_TypeOfRegistration == Initial_Secure) {

      // @siclog "Steps 22a1 - 22a16a1" siclog@//@sic R5-202596 sic@

      // IF 5GS registration type is set as Initial Registration in step 21, THEN steps 5 to 20a1 of the registration procedure described in TS 38.508-1 [4] Table 4.5.2.2-2 are performed on NR Cell 11.

      f_NR5GC_RRC_Idle_Steps5_20 (v_GMM_MobilityInfo, nr_Cell11, v_ReceivedMsg, TESTMode_OFF,-, v_TypeOfRegistration);
    }

    else if(v_TypeOfRegistration == Mobility){

      // @siclog "Steps 22b1 - 22b2a1" siclog@//@sic R5-202596 sic@

      //ELSE IF 5GS registration type is set as Mobility Registration in step 21, THEN steps 5 to 6a1 of the generic test procedure in TS 38.508-1 [4] Table 4.9.5.2.2-1 are performed on NR Cell 11.

      f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell11,v_GMM_MobilityInfo);

    }

  }
After change

function fl_TC_6_3_1_1_TestBody() runs on NR5GC_PTC

{

  var NR_SRB_COMMON_IND v_ReceivedAsp;

  var GMM_MobilityInfo_Type v_GMM_MobilityInfo;//@sic R5-202596 sic@

  var NG_NAS_MSG_Indication_Type v_ReceivedMsg;//@sic R5-202596 sic@

  var TypeOfRegistration_Type v_TypeOfRegistration;//@sic R5-202596 sic@

  var O16_Type v_Mac_UE;

  var NAS_PlmnId v_PlmnID_2 := f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (nr_Cell11));

  var O2_Type v_AccessTechId :='0800'O;//@sic R5s200663 sic@

  var octetstring v_Data := v_PlmnID_2 & v_AccessTechId;

  var template (value) SOR_Header v_Header := crs_NG_SOR_Header('1'B,'1'B,'1'B,'0'B);

  timer t_Tmax;

    //@siclog "Step 1" siclog@

  //Power on the UE

  f_UT_SwitchOnUE(UT);

    //@siclog "Steps 2-14" siclog@

  //Steps 1 to 13 of the registration procedure described in TS 38.508 subclause 4.5.2.2-2 are performed on NR Cell 13

    //@siclog "Step 15" siclog@

  //The SS transmits an DLInformationTransfer message and an REGISTRATION ACCEPT message containing

  //steering of roaming information indicating list of preferred PLMN/access technology combination provided with acknowledgment requested from the UE for successful reception

  v_Mac_UE:=f_NR5GC_SorRegistration(nr_Cell13,v_Data,v_Header);

    //@siclog "Step 16" siclog@

  //SS starts timer of tmax =(6 minutes + cell selection time) (Note 1, 2 and 3)

  t_Tmax.start(f_NR_SetTimerToleranceMax(nr_Cell13, nonProtocolTimer, 660.0));

    //@siclog "Step 17" siclog@

  //The UE transmits an ULInformationTransfer message and REGISTRATION COMPLETE message carrying acknowledgement of successful reception of the steering of roaming information

  SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell13, tsc_NR_RbId_SRB1, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

  if (not match (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg, cr_NG_REGISTRATION_COMPLETE(cr_NG_SORTransparentContainer(crs_NG_SOR_Header('0'B,'0'B,'0'B,'1'B),v_Mac_UE)))) {

f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Complete Message Failed");

}

    //@siclog "Steps 18a1-18a5a1" siclog@

  //Step 19a1 of the registration procedure described in TS 38.508-1 [4] Table 4.5.2.2-2 is performed.

    //@siclog "Step 18" siclog@

  //The SS transmits an RRCRelease message.

  f_NR5GC_RRC_Idle_Steps16_20(nr_Cell13,-,TESTMode_OFF,rrcConnectionRelease);//@sic R5-202596 sic@

    //@siclog "Step 19" siclog@

  //Check: Does the UE transmits an RRCSetupRequest on NR Cell 11 before tmax expires?

  alt {

[] SRB.check(receive(car_NR_SRB0_RrcPdu_IND(nr_Cell11, cr_38508_RRCSetupRequest(cr_NR_EstablishmentCause_Any))))

  {

t_Tmax.stop;

f_NR_PreliminaryPass (__FILE__, __LINE__, "Step 19 PASS");

  }

[] t_Tmax.timeout

  {

f_NR_SetVerdictFailOrInconc (__FILE__, __LINE__, "Step 19 Failed");

  }

  }

    //@siclog "Step 20-21" siclog@

  //Steps 3 to 4 of the registration procedure described in TS 38.508-1 subclause 4.5.2.2-2 are performed on NR Cell 11.

  //@sic R5-202596 sic@

  v_TypeOfRegistration := f_NR5GC_InitialOrMobilityRegistrationRequest (v_GMM_MobilityInfo, v_ReceivedMsg, nr_Cell11);

  if (v_TypeOfRegistration == Initial_Secure) {

      // @siclog "Steps 22a1 - 22a16a1" siclog@//@sic R5-202596 sic@

// IF 5GS registration type is set as Initial Registration in step 21, THEN steps 5 to 20a1 of the registration procedure described in TS 38.508-1 [4] Table 4.5.2.2-2 are performed on NR Cell 11.

v_Mac_UE := f_NR5GC_SorRegistration_Step5_13(nr_Cell11, v_GMM_MobilityInfo, v_ReceivedMsg, v_Data, v_Header, v_TypeOfRegistration); 

    //The UE transmits an ULInformationTransfer message and REGISTRATION COMPLETE message carrying acknowledgement of successful reception of the steering of roaming information

       SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell11, tsc_NR_RbId_SRB1, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

      if (not match (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg, cr_NG_REGISTRATION_COMPLETE(cr_NG_SORTransparentContainer(crs_NG_SOR_Header('0'B,'0'B,'0'B,'1'B),v_Mac_UE)))) {

    f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Complete Message Failed");

   }     

  //The SS transmits an RRCRelease message.

      f_NR5GC_RRC_Idle_Steps16_20(nr_Cell11,-,TESTMode_OFF,rrcConnectionRelease);//@sic R5-202596 sic@
 }

  else if(v_TypeOfRegistration == Mobility){

      // @siclog "Steps 22b1 - 22b2a1" siclog@//@sic R5-202596 sic@

//ELSE IF 5GS registration type is set as Mobility Registration in step 21, THEN steps 5 to 6a1 of the generic test procedure in TS 38.508-1 [4] Table 4.9.5.2.2-1 are performed on NR Cell 11.

f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell11,v_GMM_MobilityInfo);

  }

}
MCC160 implementation
function fl_TC_6_3_1_2_TestBody() runs on NR5GC_PTC

{

......
    //@siclog "Step 20-21" siclog@

    //Steps 3 to 4 of the registration procedure described in TS 38.508-1 subclause 4.5.2.2-2 are performed on NR Cell 11.

    //@sic R5-202596 sic@

    v_TypeOfRegistration := f_NR5GC_InitialOrMobilityRegistrationRequest (v_GMM_MobilityInfo, v_ReceivedMsg, nr_Cell11);

    if (v_TypeOfRegistration == Initial_Secure) {

      // @siclog "Steps 22a1 - 22a16a1" siclog@//@sic R5-202596 sic@

      // IF 5GS registration type is set as Initial Registration in step 21, THEN steps 5 to 20a1 of the registration procedure described in TS 38.508-1 [4] Table 4.5.2.2-2 are performed on NR Cell 11.

     f_NR5GC_SorRegistration_Step5_20(nr_Cell11,

                                     v_Data,

                                     v_Header,

                                     crs_NG_SOR_Header('0'B,'0'B,'0'B,'1'B),

                                     -,

                                     Initial_Secure,

                                     v_GMM_MobilityInfo,

                                     v_ReceivedMsg);

}
......
2.2 Change 2

	Function name
	f_NR5GC_SorRegistration_Step5_13

	Reason for change
	1. Add for change 2.1 point1. 

	Summary of change
	1. Add for change 2.1 point1.

	TTCN module
	SOR_NR5GC.ttcn

	MCC160 Comment
	Accepted in principle, but implemented differently.


New function
function f_NR5GC_SorRegistration_Step5_13(NR_CellId_Type p_NR_CellId,

                                            GMM_MobilityInfo_Type p_GMM_MobilityInfo_Type,

                                            NG_NAS_MSG_Indication_Type p_ReceivedMsg,

                                            octetstring p_Data,

                                            template (value) SOR_Header p_Header,

                                            TypeOfRegistration_Type p_RegType := Initial_Secure,

                                            boolean p_MacError:= false) runs on NR5GC_PTC return O16_Type

  {

    var O16_Type v_Mac_AMF;

    var O16_Type v_Mac_UE;

    var B128_Type v_Mac_A17;

    var B128_Type v_Mac_A18;

    var NG_NAS_SecurityParams_Type v_NAS_SecurityParams;

    var O2_Type v_Counter :='0001'O;

    var template (value) NG_NAS_DL_Message_Type v_MsgToSend;

    //Steps 5 to 13 of the registration procedure described in TS 38.508 subclause 4.5.2.2-2 are performed on NR Cell 13

    f_NR5GC_RRC_Idle_Steps5_13 (p_GMM_MobilityInfo_Type,

                                p_NR_CellId, p_ReceivedMsg,

                                f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (p_NR_CellId)),

                                TESTMode_OFF,

                                p_RegType);

    //calculate v_Mac_AMF, which is used in REGISTRATION ACCEPT message

    v_NAS_SecurityParams := f_NR5GC_Security_Get();

    v_Mac_A17 :=f_NG_Authentication_A17(bit2oct(encvalue(p_Header)),

                                  v_NAS_SecurityParams.KAUSF,

                                  v_Counter,

                                  p_Data,

                                  v_NAS_SecurityParams.KDF);

    v_Mac_AMF := bit2oct(v_Mac_A17);

    if(p_MacError)

    {

    v_Mac_AMF := not4b(v_Mac_AMF);//unmatched SOR-MAC-IAUSF

    };

    //calculate v_Mac_UE, which is used in REGISTRATION COMPLETE message after successful check

    v_Mac_A18 :=f_NG_Authentication_A18(v_NAS_SecurityParams.KAUSF,

                                  v_Counter,

                                  v_NAS_SecurityParams.KDF);

    v_Mac_UE :=bit2oct(v_Mac_A18);

    //The SS transmits an DLInformationTransfer message and an REGISTRATION ACCEPT message containing expected steering of roaming information

    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Initial_Secure,

                                                   f_NR5GC_GetMobileIdGUTI (p_NR_CellId),

                                                   cs_NG_TAIListNonConsecutive(f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (p_NR_CellId)),

                                                                                {bit2oct(f_NR_CellInfo_GetTAC(p_NR_CellId))}),

                                                    -,-,-,-,-,-,-,-,-,-,-,-,-,-,-,-,-,

                                                    cs_NG_SORTransparentContainer_IeiIel(p_Header, v_Mac_AMF, v_Counter, p_Data));

    // Registration Accept

    SRB.send(cas_NR_SRB_NasPdu_REQ(p_NR_CellId,

                                   tsc_NR_RbId_SRB1, // must be on SRB1

                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));

    return v_Mac_UE;

  }
MCC160 implementation
function f_NR5GC_SorRegistration_Step5_20(NR_CellId_Type p_NR_CellId,

                                            octetstring p_Data,

                                            template (value) SOR_Header p_Header,

                                            template (value) SOR_Header p_Header_UE:=omit,

                                            boolean p_MacError:= false,

                                            TypeOfRegistration_Type p_RegType := Initial_NoSecurity,

                                            GMM_MobilityInfo_Type v_GMM_MobilityInfo,

                                            NG_NAS_MSG_Indication_Type v_ReceivedMsg,

                                            boolean p_AckSor:= true) runs on NR5GC_PTC 

  {

    var NR_SRB_COMMON_IND v_ReceivedAsp;

    var O16_Type v_Mac_AMF;

    var O16_Type v_Mac_UE;

    var B128_Type v_Mac_A17;

    var B128_Type v_Mac_A18;

    var NG_NAS_SecurityParams_Type v_NAS_SecurityParams;

    var O2_Type v_Counter :='0001'O;

    var template (value) NG_NAS_DL_Message_Type v_MsgToSend;

    f_NR5GC_RRC_Idle_Steps5_13 (v_GMM_MobilityInfo,

                                p_NR_CellId, v_ReceivedMsg,

                                f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (p_NR_CellId)),

                                TESTMode_OFF,

                                p_RegType);

    //calculate v_Mac_AMF, which is used in REGISTRATION ACCEPT message

    v_NAS_SecurityParams := f_NR5GC_Security_Get();

    v_Mac_A17 :=f_NG_Authentication_A17(bit2oct(encvalue(p_Header)),

                                  v_NAS_SecurityParams.KAUSF,

                                  v_Counter,

                                  p_Data,

                                  v_NAS_SecurityParams.KDF);

    v_Mac_AMF := bit2oct(v_Mac_A17);

    if(p_MacError)

    {

    v_Mac_AMF := not4b(v_Mac_AMF);//unmatched SOR-MAC-IAUSF

    };

    //calculate v_Mac_UE, which is used in REGISTRATION COMPLETE message after successful check

    v_Mac_A18 :=f_NG_Authentication_A18(v_NAS_SecurityParams.KAUSF,

                                  v_Counter,

                                  v_NAS_SecurityParams.KDF);

    v_Mac_UE :=bit2oct(v_Mac_A18);

    //The SS transmits an DLInformationTransfer message and an REGISTRATION ACCEPT message containing expected steering of roaming information

    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Initial_Secure,

                                                   f_NR5GC_GetMobileIdGUTI (p_NR_CellId),

                                                   cs_NG_TAIListNonConsecutive(f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (p_NR_CellId)),

                                                                                {bit2oct(f_NR_CellInfo_GetTAC(p_NR_CellId))}),

                                                    -,-,-,-,-,-,-,-,-,-,-,-,-,-,-,-,-,

                                                    cs_NG_SORTransparentContainer_IeiIel(p_Header, v_Mac_AMF, v_Counter, p_Data));

    // Registration Accept

    SRB.send(cas_NR_SRB_NasPdu_REQ(p_NR_CellId,

                                   tsc_NR_RbId_SRB1, // must be on SRB1

                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));

    if(p_AckSor)

    {

    SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, tsc_NR_RbId_SRB1, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

      if (not match (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg, cr_NG_REGISTRATION_COMPLETE(cr_NG_SORTransparentContainer(p_Header_UE,v_Mac_UE)))) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Complete Message Failed");

      }

      f_NR5GC_RRC_Idle_Steps16_20 (p_NR_CellId,-, TESTMode_OFF,rrcConnectionRelease);

     }

    else{

      f_NR5GC_RRC_Idle_Steps15_20(p_NR_CellId, TESTMode_OFF, rrcConnectionRelease);

    }

  }  
